
はじめに 

 

本書は、SystemVerilog Class Library (SCL) の解説書です。SCL は小規模な検証パッケージなの

で、全体を理解する事は容易です。更に、SCL の軽量性はコンパイルの高速化と実行モジュ

ールサイズの小型化に通じ、TAT の短縮効果があります。 

 

UVM は SystemVerilog で記述された優れた検証パッケージですが、UVM の難易度が高い事も

事実です。このため、多くの学習時間が初期コストとして発生し、UVM を十分に駆使する事

ができる状況に至るまでの道のりが遠く見えるのが現状です。一方、膨大な行数からなる

UVM の解読は殆ど不可能に近いだけでなく、基本的な機能に関しても多くのミステリーが存

在します。UVM には深い理由があり現在のような複雑な実装になっていると考えられますが、

高度な機能を必要としない検証処理では、むしろ簡潔明瞭な処理方式の方が望ましいと言え

ます。 

 

SCL は、簡潔明瞭で誰でも短期間に理解できる検証ライブラリーの構造を備えています。例

えば、UVM のクラス階層と異なり、SCL は理路整然としたクラス階層を実現しています。更

に、UVM のトランザクション処理では、重要な処理部分が表面から隠されているため、殆ど

の人が理解し得ない難攻不落の要塞になっています。UVM を開発した者しか理解し得ないよ

うな処理方式は一般的なライブラリーとしては不適切です。SCL は、その様な設計思想を排

除し、誰にでも分かり易い処理方式を採用しています。具体的には、SCL には以下のような

特徴があります。 

 

 UVM のクラス階層と異なり、理路整然としたクラス階層を実現。 

 マクロを使用し記述を簡単に記述できるように入力の省力化。 

 TLM ポートを標準装備したコレクタークラスを追加。 

 モニタークラスには標準的な TLM ポートを定義。 

 トランザクション処理には UVM とは異なる方式を採用し、トランザクションの流れを

可視化。 

 virtual インターフェースの使用手順を簡素化。 

 検証作業に重宝する汎用的な機能とユーティリティを追加。 

 

例えば、virtual インターフェースの使用法が如何に簡単であるかを示します。先ず、インタ

ーフェースを定義した時に、以下のように SCL マクロを使用して virtual インターフェース

を操作するための vif_config クラスを生成します。 

 

interface simple_if; 

... 

endinterface 

`scl_vif_config_m(virtual simple_if,vif_config) 

 

こうすると、トップモジュールではインターフェースのインスタンスを以下のように設定で

きます。 

 

vif_config::set(SIF); 

 

そして、検証コンポーネントでは、以下に示すような簡単な手順で virtualインターフェース

を取得できます。こうする事により、検証コンポーネント内でインターフェース名をハード

コーディングしないで済みます。 

 

vif = vif_config::get(); 

もう一つ別の例を紹介します。UVM のトランザクショ

ン生成手順は非常に複雑であり理解し難いという欠点が

あります。その欠点を克服するために、SCL ではトラン

ザクションの生成手順をツリーで表現します。ツリーの

ルートはplayと呼ばれるオブジェクトで、他のノードは

act と呼ばれるオブジェクトで表現されます。ツリーの

内部ノードはトランザクションの生成手順を示し、リー

フノードは一つのトランザクションを生成する役目を持

ちます。テストがシナリオを決定するとジェネレータに

play が割り当てられて右図のようなツリーが構築されて

トランザクションを生成するための手順が確定します。

ドライバーがジェネレータにトランザクション取得要求を発行するとリーフノードが呼ばれ

トランザクションが生成されてドライバーに戻されます。赤いリーフノードは、次にトラン

ザクションを作る役目を持つノードを示しています。赤いリーフノードの処理が終わるとジ

ェネレータは右隣のリーフノードを次の処理対象に設定します。このようにトランザクショ

ンとリーフノードが対応しているので、トランザクション生成の仕組みは理解し易くなって

います。 

 

SCL では、一般的に普及されている用語を踏襲し、SCL 独自のアーキテクチャに依存する部

分には SCL 固有の用語を使用します。例えば、トランザクションを処理するモデリングには

一般的な TLM という用語を使用します。一方、トランザクション処理を制御する機能にはシ

ナリオという用語を使用します。SCL は UVM とはトランザクション処理が異なるので、混

乱を避けるために UVM と同じ用語を使用するのを避けています。例えば、SCL では UVM で

使用されているシーケンスやシーケンサーという用語を使用していません。また、検証の対

象である DUT は RTL で記述されていると仮定します。 

 

本書は、幾つかの章から構成されていますが、第 1 章が最も重要な内容になっています。内

容的には SCL の概要的な解説であるため、他の章を読むまでは理解できない部分もあると思

います。既に触れたように、SCLの特徴は virtualインターフェースの使用手順とトランザク

ション生成処理にあるので、第 3 章、第 4 章、第 8 章は非常に重要な意味を持ちます。これ

らの章は丁寧にお読みください。第 9 章では、SCL を使用して検証環境を構築する例を紹介

しているので、必要に応じて参照して下さい。 

 

SCL には、トランザクションや検証コンポーネントの他に検証作業で必要になる汎用的な機

能も含まれています。例えば、パターンマッチング機能、文字列操作機能、プリント書式機

能、クロック生成機能、プロセス生成機能、プライオリティキュー、ファイル入出力等の機

能も SCL に含まれています。特に、SCL は強力なプリント機能を備えているので、殆どのプ

リント処理は省力化されます。例えば、プリント用のレイアウトをヘッダと呼ばれるデータ

構造で定義すると、カラムに表示する値を設定するだけで、カラムの自動調節が行われます。

そして、2 進や 16 進の表現形式の変更は使用するマクロ名を変更するだけで済みます。また、

ビット幅を変更しても自動的にカラム調節が行われます。第 10 章はこれらの便利な機能を詳

細な使用例と共に解説をしています。 

 

尚、紙面の都合上、一部の記述は小さな書体で記述されています。ただし、書体が小さいか

らと言って内容の重要性が低いわけではありません。 

 

アートグラフィックス 

 

 

play 

act 

変更履歴 

 

日付 Revision 変更点 

2023.12.10 1.0 初版。 

2024.01.08 1.1 
 パターンマッチング機能（正則表現機能）を追加。 

 コンフィギュレーション設定変更機能を追加。 

2024.01.25 1.2 

 オブジェクトとコンポーネントにコピー機能を追加し

たので、トランザクションのコピー等が簡単にできよ

うになりました。 

 プリント書式の自動カラム調節機能を追加しました。 

 スコアボードに predict()メソッドを追加しまし

た。 

 スコアボードを使用して検証環境を構築する例を追加

しました。 

2024.02.05 1.3 

 ノンブロッキング方式によるトランザクション送受信

するためのポート機能を追加。 

 トランザクションをプーリングする FIFO 機能を追

加。 

1 概要 

SCL は TLM（Transaction Level Modeling）による検証環境を構築するための SystemVerilog ク

ラスライブラリーです。このライブラリーには、検証環境を構築するために必要な検証コン

ポーネントのベースクラス、トランザクションを定義するためのベースクラス、トランザク

ションを操作するためのメソッド、プリント機能、文字列操作機能、ファイル入出力機能、

パターンマッチング機能、ユーティリティ等が含まれています。本章では、SCL の概要を解

説します。 

1.1 SCL の意義と目的 

検証を行う際には、検証の対象である DUT に対して、検証環境の構築、検証用のデータ設計、

検証結果の集計とレポート方式等を決定しなければなりません。このうち、検証用のデータ

設計と検証結果の集計法は、明らかに、DUTに依存しますが、検証環境の構築にはDUTの機

能にかかわらずほぼ普遍的な機能が必要になります。 

 

例えば、対象の DUT がどのような機能を持つにしても、検証環境にはドライバーの機能が必

要になります。同様に、どのような検証においても、DUT からのレスポンスをサンプリング

する機能を持つコレクター機能が必要になります。更に、ドライバーに検証用のデータを供

給するジェネレータ、およびコレクターが収集したデータを解析するためのモニター、スコ

アボード等の機能も必要になります。しかも、これらの機能は DUT の種類にかかわらず、殆

ど全ての検証環境で必要になります。したがって、これらの不可欠な機能を DUT に依存しな

い形式で備えておけば、検証環境構築作業の生産性を著しく向上できます。SCL は、その様

な普遍的な機能を供給する SystemVerilog クラスライブラリーです。 

1.2 検証環境を構成する概念 

SCL では、検証に用いるデータをトランザクションと呼びます。トランザクションは、RTL

の信号（すなわち、ネット）よりも高位な概念で、タスクやファンクション等のメソッドを

使用してトランザクションを操作します。一方、DUT は RTL信号の流れで機能が表現されて

います。トランザクションと信号では、全く異なる次元のデータ表現であるため、相互変換

が必要になります。それが、ドライバーとコレクターの役目です。 

 

ドライバーは、SCL の検証コンポーネントにより生成されたトランザクションを RTL 信号に

変換してDUTをドライブします。一方、コレクターは、DUTからサンプリングしたレスポン

スをトランザクションに逆変換をして他の検証コンポーネントに転送します。ドライバーが

DUTをドライブする時、およびコレクターがDUTからレスポンスをサンプリングする時に、

SystemVerilog の virtual インターフェース（以降、変数名で表現する際には vif と略称しま

す）が使用されます（図 1-1）。 

 

検証過程はトランザクションを使用して行われるので TLM と呼ばれ RTL と明確に区別され

ます。DUT はインターフェースを使用する必要はありませんが、インターフェースには DUT

のポートに対応する信号が定義されています。したがって、DUT は仮想的にインターフェー

スを使用していると考えられます。図 1-1 は、その様に拡張解釈した概念を基にしています。 

 

 

図 1-1 検証環境でのドライバー、コレクターと DUT の関連 

 

collector 

driver 

1.3 インターフェースと virtual インターフェース 

インターフェースは、通常、トップモジュールに配置されて以下のように DUT と接続されて

います。 

 

interface simple_if(input logic clk); 

... 

endinterface 

 

module top; 

bit  clk; 

 

simple_if SIF(.*); 

device DUT(.clk(SIF.clk),...); 

... 

endmodule 

 

DUT はインターフェースを使用していなくても、インターフェース内の信号に接続されてい

れば、DUT は仮想的にインターフェースとつながれていると考えられます。即ち、どのよう

な DUT に対しても virtual インターフェースを適用できる機会が存在します。 

 

そして、トップモジュールはドライバーやコレクターに定義されている vif にインターフェ

ースのインスタンスを割り当てる準備をします。こうして、検証環境では virtual インターフ

ェースを使用できるようになります（図 1-2）。 

 

 

図 1-2 トップモジュールに配置されたインターフェース、DUT、検証環境の関連 

 

vif はインターフェースのインスタンスへのポインターを表現しているので、ドライバーや

コレクターがあたかも DUT と直接接続されているような状態を表現できます（図 1-1）。つ

まり、ドライバーが vif を介して信号を操作すると、DUTの対応する信号が反応します。同

様に、DUTの信号に変化が起これば、コレクターは vif を介してその変化を知る事ができま

す。 

1.4 virtual インターフェースの操作 

virtual インターフェースは、SystemVerilog が備えている機能です。SystemVerilog のクラス内

にはインターフェースのインスタンスを配置できないため、virtual インターフェースを使用

しなければなりません。virtual インターフェースとは、インターフェースのインスタンスへ

のポインターを示す変数で、その変数にインターフェースのインスタンスが割り当てられな

ければなりません。 

 

virtual インターフェースをクラス内に定義する際には、インターフェース名称が必要になり

ますが、インターフェース名称を検証環境で直接参照すると、検証環境自身の汎用性が失わ

れる恐れがあります。そのため、SCL ではインターフェースを定義する際に virtual インター

フェースを操作するためのクラスを定義する仕組みになっています。この仕組みによりクラ

ス内でインターフェース名を参照する事を回避できます。例えば、インターフェースを定義

した際に、以下のようにして virtual インターフェースを操作するためのクラス vif_vonfig

も定義しておきます。 

 

top 

DUT SIF 検証環境 

vif 

DUT はインターフェースに定義さ

れている信号に接続されている 

interface simple_if(input logic clk); 

... 

endinterface 

 

`scl_vif_config_m(virtual simple_if,vif_config) 

 

こうすると、トップモジュールで virtual インターフェースの設定準備が可能になります。ま

た、ドライバーやコレクターは、トップモジュールにより設定された virtual インターフェー

スを vif_config クラスにより取得できます。 

 

図 1-3 では vif_config を使用してトップモジュールが virtual インターフェースを設定して

いる様子を示しています。そして、設定された virtual インターフェースをドライバーが取得

しています。こうして、検証環境でインターフェース名をハードコーディングする状況から

回避できます。 

 

 

図 1-3 インターフェースのインスタンスと virtual インターフェース 

 

SCL は virtual インターフェースを設定するための便利なメソッドを提供しています。第 4 章

で、virtual インターフェースの設定法を詳しく解説します。この方法は、SCL に限らず一般

的に適用できる便利な手法なので、確実に理解をする事をすすめます。 

1.5 ダイナミックな検証環境 

検証コンポーネントは virtual インターフェースを保有してさえいれば、DUT と通信を行える

ので、検証コンポーネントは、スタテックなインスタンスとして存在する必要はありません。

つまり、virtual インターフェースを使用する事により、ダイナミックに検証環境の構成を変

更できる事になります。換言すれば、複数のテストケースを用意して、実行時にテストケー

スを決定できる事になり、検証作業の生産性が向上します。つまり、一度のコンパイルで複

数のテストケースを実行できるようになります。このように、SCL は検証コンポーネントを

記述するための機能を提供するだけでなく、検証作業全体の生産性を促進する効果をもたら

します。 

1.6 ドライバーとトランザクション 

さて、ドライバーは DUT の機能に関わらず必要な存在ですが、ドライバーの処理内容は、

DUTに依存します。したがって、ドライバーにはDUTの機能に応じて変化する仕組みが必要

になります。DUT をドライブするためには、以下の機能が必要になります。 

 

 DUT の信号を操作するために必要なデータ 

 DUT の信号を操作する手順 

 

前者のデータはトランザクションと呼ばれ、後者の手順はメソッドと呼ばれます。メソッド

は、ドライバイー内に SystemVerilog のタスク、およびファンクションとして定義されます。

トランザクションには、一定の規則が必要であるため、SCL ではトランザクションのベース

クラスを定義しています。そして、ユーザがトランザクションを定義する際には、そのベー

module top; 

bit  clk; 

  

simple_if SIF(.*); 

device DUT(SIF); 

... 

initial begin 

 vif_config::set(SIF); 

 ... 

end 

endmodule 

 

class driver_t ...; 

vif_config::vif_type vif; 

... 

function void connect_phase(...); 

 super.connect_phase(param); 

 vif = vif_config::get(); 

endfunction 

endclass 

スクラスのサブクラスとして定義しなければなりません。同様に、SCL ではドライバーのベ

ースクラスを定義しています。ユーザのドライバーは、直接または間接的にこのベースクラ

スのサブクラスとして定義されなければなりません。 

 

ドライバーは、DUT の機能に応じて変化しなければならないので、ドライバーのベースクラ

スには DUT の機能を表現するパラメータが必要になります。そのパラメータは、トランザク

ションです。したがって、ドライバーのベースクラスは以下のように定義されています。こ

こで、scl_component_t は SCL における全ての検証コンポーネントのベースクラスです。 

 

class scl_driver_t 

 #(type TR=scl_transaction_t) extends scl_component_t; 

... 

endclass 

 

ドライバーと同様にトランザクションのベースクラスが定義されていますが、以下に示すよ

うに、特別なパラメータを伴いません。この点は、ドライバーと大きく異なります。前述し

たように、ユーザがトランザクションを定義する際には、直接または間接的にこのベースク

ラスのサブクラスとして定義しなければなりません。ここで、scl_object_tは SCLにおけ

る全てのオブジェクトのベースクラスです。 

 

class scl_transaction_t extends scl_object_t; 

... 

endclass 

 

1.7 トランザクションの操作 

トランザクションの操作は、TLM の手法に従います。RTL では、モジュールのポート同士が

ネットまたは変数で接続されて信号値が常に流れている状態を表現します（図 1-4）。 

 

 

図 1-4 RTL による接続方式 

一方、検証コンポーネントは DUT と物理的に接続されていないので、TLM では RTL とは全

く異なる処理形態が必要になります。TLM はデータの送受信をメソッドの呼び出しで行いま

す。例えば、トランザクションを取得するには get()メソッドを使用し、トランザクション

を送るためには、put()メソッドを使用します。 

1.7.1 トランザクションを取得する手順 

トランザクションを get()メソッドで取得するためには、トランザクションを準備する側が

get()メソッドの内容を実装しなければなりません。図 1-5 においては、get_server は

get()メソッドの処理内容を定義しなければなりません。 

 

 

図 1-5 get()メソッドによるトランザクションの取得の流れ 

module-2 module-1 
a 

b 

a 

b 

get_client get_server 

get(item) 

item 

通常はこの様なプロトコルにしたがってトランザクションを取得しなければなりませんが、

SCL が提供するドライバーとジェネレータを使用してクラスを定義する事により、低レベル

の get()や put()メソッドの使用から解放されます。 

1.7.2 トランザクションを送信する手順 

トランザクションの取得と同様に、トランザクションを put()メソッドで引き渡すためには、

受け取り側は put()メソッドの内容を実装しなければなりません。図 1-6 においては、

put_server は put()メソッドの処理内容を定義しなければなりません。 

 

 

図 1-6 put()メソッドによるトランザクションの送信の流れ 

 

1.7.3 トランザクションを複数の検証コンポーネントに送信する手順 

これまでに紹介した put()と get()メソッドは、トランザクションを能動的に処理する場合

に使用するTLMですが、受動的（つまり、reactive）にトランザクションを処理する場合には

write()メソッドを使用します。 

 

例えば、DUT からのレスポンスをサンプリングする事により発生するトランザクション処理

は受動的なので、write()メソッドを使用します。図 1-7 は、コレクターがモニターに

write()メソッドでトランザクションを送信している状態を示しています。DUT からのレス

ポンスは、コレクターにより常に監視されなければならないため、コレクターは write()メ

ソッドに時間を消費できません。このため、write()メソッドはファンクションとして実装

されます。モニターはwrite()メソッドを実装しなければなりません。尚、write()メソッ

ドは即座に戻るため、トランザクションを複数の検証コンポーネントに一斉に送信できる利

点があります。 

 

 

図 1-7 write()メソッドによるトランザクションの送信 

 

以上述べたように、TLM ではトランザクション処理メソッドの呼ばれる側がトランザクショ

ンの処理方式を決定する事になります。更に詳しいトランザクション処理手順は第 3 章で解

説します。 

1.8 トランザクションの生成 

トランザクションは scl_transation_t と呼ばれるクラスのサブクラスとして表現されま

すが、SCL ではトランザクションの生成および準備を scl_play_t と scl_act_t と呼ばれ

るクラスで行います。それらのクラスはトランザクションの生成手順をツリーで表現するた

めに使用されます。 

 

ツリーのルートを scl_play_t のオブジェクトで表現し、ツリー内の他のノードを

put_client put_server 

put(item) 

collector monitor 

write(item) 

scl_act_t のオブジェクトで表現します（図 1-8）。そして、ツリーのリーフノードは一つ

のトランザクションを生成する役目を持ちます。他のノードは、下位のノードを使用してト

ランザクションの生成手順を制御します。ツリー構造には制限がないため複雑なトランザク

ション生成手順を表現できます。scl_act_tにはget_item()とget_group()メソッドが

あり、get_item()を呼ぶと一つのトランザクションを生成でき、get_group()を呼ぶと

他の scl_act_t のオブジェクトを呼び出せます。つまり、階層を表現できます。 

 

 

図 1-8 トランザクション生成ツリー 

 

ドライバーがジェネレータにトランザクションを要求すると、ジェネレータは処理対象のリ

ーフノード（図 1-8 における赤いノード）を実行してトランザクションを作りドライバーに

引き渡します。赤いリーフノードの処理が終了すると右隣のリーフノードが次に処理される

対象となり赤くなります。この操作を繰り返してツリーのリーフノードを左から順に処理し

て一連のトランザクションが生成されます。生成手順の中にループを記述できるので同じ手

順を複数回繰り返す事もできます。したがって、簡潔にトランザクション生成手順を実現す

る事ができます。 

1.9 テストとシナリオ 

SCL では、テストがシナリオを割り当てる役目を持ちます。ここで、シナリオとはトランザ

クション生成手順を意味します。テストがシナリオを割り当てると、シナリオはジェネレー

タに scl_play_t のオブジェクトを割り当ててトランザクション生成ツリーを作る準備を開

始させます。ただし、このツリーは最初から完成した構造ではなくドライバーがトランザク

ションを取得して行く過程において次第に構造を完成して行きます。 

 

シミュレーションが開始すると、ジェネレータはトランザクション生成ツリーのルートノー

ドを起動してトランザクション生成処理の初期化を行います。その後は、ドライバーからト

ランザクションの要求が来るのを待ちます。要求が来ると前節で解説したようにトランザク

ション生成手順が作動します。 

1.10 検証環境の構成 

テストベンチ1は、所謂、トップモジュールを指します。テストベンチには、DUTとインター

フェースのインスタンスが配置されるだけですが、SCL のクラスを使用して開発されたテス

トがテストベンチにより選択されて実行されます。テストベンチの構成は検証内容に依存す

る事は明らかですが、一般的には、図 1-9 のような構成になります。それらの構成要素の解

説は表 1-1 にあります。検証環境を構成するための SCL クラスはメソドロジークラスとも呼

ばれます。 

 

テストは、virtualインターフェースを介して DUTをドライブする事、および DUTからのレス

ポンスを収集して DUT の動作を検証および解析する役目を担当します。テストベンチには幾

つかのテストが含まれますが、その中の一つが選択されて実行します。 

 
1 テストベンチの定義は人により異なりますが、ここではトップモジュールと同意語としま

す。 

scl_play_t のオブジェクト 

scl_act_t のオブジェクト 

scl_act_t のオブジェクト 

エージェントは最も下位にある階層コンポーネントで DUT をドライブする機能と DUT から

のレスポンスをサンプリングする機能を持ちます。エージェントにはジェネレータ、ドライ

バー、コレクター、モニター等のコンポーネントが配置されます。 

 

 

図 1-9 テストベンチの構成 

 

表 1-1 テストベンチを構成する要素 

階層を構成する要素 意味 

テストベンチ 

これは、いわゆる、トップレベルのモジュールを指します。このテ

ストベンチは DUT およびインターフェースをインスタンスとして

保有します。検証環境を決定して起動する役目も持ちます。 

テスト 
検証用のテストケースのトップレベルのクラスインスタンスを意味

します。このインスタンスが該当するテストを実行します。 

エンバイロンメント 階層的に構築した検証コンポーネントです。 

スコアボード 

スコアボードの主目的は、DUT の動作を確認する事です。すなわ

ち、エージェントから取得した DUT のレスポンスが期待される結

果と一致するかのチェックを遂行します。 

エージェント 

基本的な検証コンポーネントで、ジェネレータ、ドライバー、モニ

ター、コレクター等のインスタンスで構成されます。エージェント

は DUT と virtual インターフェースを介して接続されます。 

ジェネレータ 

トランザクションを生成する制御を司るコンポーネントです。ジェ

ネレータはアクトを作り実行する事によりトランザクションを生成

します。 

 

エンバイロンメント 

テスト 

テスト 

テストベンチ 

DUT 

エンバイロン

メント 

エンバイロン

メント 

エージェント 

エージェント 

スコアボード 

ドライバー 

ドライバーはジェネレータからトランザクションを取得し、シグナ

ルレベルに変換した後、DUT をドライブします。ジェネレータと

ドライバーの通信には TLM が使用されます。一方、DUT をドライ

ブするためには、virtual インターフェースが使用されます。 

コレクター 

DUT からのレスポンスをサンプリングしてトランザクションに変

換し、他の検証コンポーネントに送信する役目を持ちます。つま

り、コレクターは RTL より TLM への変換機能です。 

モニター 

モニターはコレクターから受信したトランザクションを他の検証コ

ンポーネントに一斉に送信します。モニター自身もカバレッジ計

算、およびレスポンスに関する簡単なチェックも行います。 

アクト 

トランザクションを生成するための手順を含むオブジェクトです。

オブジェクトであるため、コンポーネント階層には含まれません。

アクトは他のアクトを呼び出す事ができるので複雑なテストシナリ

オを定義する事ができます。 

プレイ 

トランザクションを生成するための手順はツリー状に配置されたア

クトのオブジェクトで表現されます。ツリーのルートはプレイのオ

ブジェクトで他のツリーノードと区別されます。プレイはジェネレ

ータに割り当てられるとツリーを構築する準備を開始します。 

シナリオ 検証環境にトランザクション生成手順を割り当てる働きをします。 

 

1.11 ダイナミックなコンフィギュレーション設定変更 

検証中にはコンフィギュレーションを一時的に変更する必要性が出てきます。SCL では、該

当するクラスを修正せずにクラスのプロパーティを変更できます。また、クラス自身を他の

サブクラスに置き換える事さえ可能です。例えば、検証環境にドライバー（driver_t）のイン

スタンスが使用されているとすると、検証環境を修正せずに driver_t と互換性のあるクラス

new_driver_t のインスタンスで置き換える事ができます（図 1-10）。 

 

 

図 1-10 実行時のコンフィギュレーション設定変更 

 

同様に、検証環境の外部からクラスのプロパーティの設定変更ができます。したがって、検

証環境を修正せずに一時的に新しい機能の確認をする事ができます。この便利な機能は第 2

章で解説されます。 

1.12 SCL マクロ 

SCL では、同じ内容を持つ命令を頻繁に記述する場合、マクロを使用するようにしています。

基本的には、マクロはタイプ入力の負荷を減少させる効果を持ちますが、多くの複雑な命令

群を生成するためにも使用されます。マクロの簡単な使用例としてドライバーの定義を以下

に紹介します。 

 

driver_t 

インスタンス 

new_driver_t 

インスタンス 

検証環境 検証環境 

2 SCL の構成 

SCL のクラスは、オブジェクト系と検証コンポーネント系に分類されますが、

scl_object_t と scl_component_t がそれぞれのベースクラスとなっています。つまり、

全てのオブジェクトは、直接または間接的にscl_object_t のサブクラスで、全ての検証コ

ンポーネントは、直接または間接的に scl_component_t のサブクラスです。

scl_object_t と scl_component_t クラスをユーザが直接使用する機会は少ないと思い

ますが、これらのクラスが備えている機能を理解しておく必要があります。何故なら、それ

らの機能の多くは virtual メソッドとして定義されているからです。 

2.1 クラス階層 

SCL では、ごく自然で、かつ現代的な概念に基づいてクラスの設計がされています。SCL の

ベースクラスは、scl_thing_t であり、全てがそこから始まります。トランザクションも

ポートもコンポーネントも「物」であるので、その概念は適合します。scl_thing_t から

派生したクラスは、物を具体的に表現し始める目的を持つため、SCL ではアプリケーション

に適合する命名法を使用しています。 

 

例えば、シナリオはテストケースに対応し、シナリオはプレイを実行させます。プレイは、

トランザクションを作るためにアクトを使います。図 2-1は、SCLのクラス階層の一部です。

クラス階層に現れるクラスの概要を表 2-1 に示します。SCL のクラスは、検証コンポーネン

ト系とオブジェクト系に分類され、オブジェクト系のクラスは検証コンポーネントが使用す

る機能を提供します。 

 

 

図 2-1 SCL クラス階層（抜粋） 

 

表 2-1 主な SCL クラスの概要 

SCL クラス 機能 

scl_thing_t 
他の全ての SCLクラスのベースクラスで、アブストラクトクラスとし

て定義されています。 

scl_component_t 全ての検証コンポーネントのベースクラスです。 

scl_object_t 全てのオブジェクトのベースクラスです。 

scl_port_t TLM ポートのベースクラスです。 

scl_scenario_t シナリオのベースクラスです。 

scl_play_t トランザクションを作る手順のベースクラスです。 

scl_act_t トランザクションを作るためのベースクラスです。 

scl_transaction_t トランザクションを定義するためのベースクラスです。 

 

これらのクラスの機能は、第 3 章以降で詳しく解説されます。 

2.2 scl_void_t 

実は、SCL にはもう一つ別のクラス scl_void_t  が定義されていて scl_thing_t のベー

スクラスになっていますが、通常使用される事はないので scl_thing_t が全てのベースク

ラスであるという表現をしています。 

thing 

component object 

port 

scenario play act transaction 

次節に scl_thing_t の解説がありますが、コンストラクタはパラメータとして name を必

要とします。一方、一般的な記述用としてパラメータを必要としないクラスも必要になるた

め、scl_void_t が用意されています。 このクラスの存在により、どのようなクラスも

scl_void_t クラスのサブクラスになるように定義できます。要約すれば、scl_void_t ク

ラスは SCL の完全性を保証する目的に存在し、scl_thing_t は SCL の実質的なベースクラ

スです。因みに、scl_void_t は以下のように定義されています。 

 

virtual class scl_void_t; 

endclass 

 

2.3 scl_thing_t 

このクラスは、全ての SCL クラスのベースクラスで、このクラス自身のインスタンスを作る

意義がないので、アブストラクトクラスとして定義されています。scl_thing_t の全容は、

以下のようになります。 

 

virtual class scl_thing_t extends scl_void_t; 

// members 

static int m_count; // # of instances created 

string   m_name; 

local int  m_id;  // unique number in entire environment 

const static string m_type_name = "scl_thing_t"; 

 

// methods 

extern function new(string name); 

extern virtual function string get_name(); 

extern virtual function int get_id(); 

extern virtual function string get_type_name(); 

extern virtual function bit is_object(); 

extern virtual function bit is_component(); 

// the following function is implemented by macros 

extern virtual function void 

 scl_field_processor(input scl_thing_t source,string tabs="", 

  scl_field_command_e command=SCL_FIELD_PRINT); 

pure virtual function void print(); 

pure virtual function void copy(scl_thing_t source); 

endclass 

 

このクラスに定義されているプロパーティを表 2-2 に紹介します。 

 

表 2-2 scl_thing_t のプロパーティ 

プロパーティ 説明 

m_count 

生成されたインスタンスの総数を管理します。このプロパーティ

を利用して、全てのインスタンスにユニークな番号（ID）を割り

付けます。 

m_name scl_thing_t のインスタンスに付けられた名称を示します。 

m_id scl_thing_tのインスタンスに割り当てられたユニークな IDです。 

m_type_name 
クラスのタイプを示す文字列です。サブクラスにより再定義され

ます。 
 

このクラスに定義されているメソッドの機能概要を表 2-3 に示します。 

3 TLM 

SCL では、トランザクションの取得に get()、送信2に put()、転送に write()メソッドを

使用します。何れのメソッドを使用する際にも、クライアントとサーバーがあります。クラ

イアントは、命令を発行する側で、その命令を処理する側がサーバーです。そして、サーバ

ーが処理手順をメソッドとして定義します。送信と転送命令は、ポートを介して行われます。

SCL では、この関係を図 3-1 のように表現します。サーバー側は、メソッドの内容を埋める

という意味でポートを黒く塗りつぶします。 

 

 

図 3-1 SCL でのクライアントとサーバーの表現法 

 

クライアントとサーバーにはポートが配置されますが、それらのポートの接続は、クライア

ントとサーバーのインスタンスを保有する検証コンポーネントが行います。したがって、ク

ライアントは特別なサーバーを仮定していません。同様に、サーバーは特別なクライアント

を前提としていません。メソッドで指定するプロトコルに互換性がある限り、クライアント

とサーバーの通信は正しく動作します。 

 

本章では、以下のように定義されたトランザクションを使用して、トランザクションを操作

するメソッドの定義法と使用法を順に解説します。 

 

class simple_item_t extends scl_transaction_t; 

rand logic [3:0] a, 

     b; 

    

function new(string name="simple_item"); 

 super.new(name); 

endfunction 

 

endclass 

 

本章では、トランザクションの送受信方式を解説しますが、メソドロジークラスを使用すれ

ば、ここで解説されている細かい物理的な処理手順から解放されます。したがって、メソド

ロジークラスを使えば本章で解説する内容の詳細を理解する必要はありませんが、TLM ポー

トの概念は重要なので理解しておくと便利です。例えば、ユーザ自身で機能を拡張する際に

は重要な知識となります。 

3.1 ポート 

SCL では、ポートを介してトランザクションの授受を行います。一対一のポート授受には、

scl_io_port_t ま た は scl_nbio_port_t を 使 用 し 、 一 対 N の 送 信 に は

scl_pass_port_t を使用します。ポートはクラスとして表現され、図 3-2 に示すようなク

ラス階層を構成します。それぞれのクラスの機能概要は表 3-1 に示されています。図中に現

れる接頭辞 nb は non-blockingの略称です。つまり、ノンブロッキング方式で送受信しま

す。 

 

 
2 通信の対象が一つである場合には送信、複数であれば転送と表現しています。 

クライアント サーバー 

図 3-2 SCL ポートクラスの階層構造 

 

表 3-1 SCL のポートクラス 

ポートクラス 機能 

scl_port_t 
全ての SCL ポートのベースクラスです。全てのポートに共通する情

報を定義しています。 

scl_io_port_t 一対一のトランザクション処理をするポートのベースクラスです。 

scl_get_port_t トランザクションを取得するポートのベースクラスです。 

scl_get_server_t トランザクションを準備するポートのベースクラスです。 

scl_put_port_t トランザクションを送信するポートのベースクラスです。 

scl_put_server_t 
送信されたトランザクションを処理するポートのベースクラスで

す。 

scl_nbio_port_t 
一対一のトランザクション処理をするポートのベースクラスです

が、ノンブロッキング方式を用います。 

scl_nbget_port_t 
ノンブロッキング方式でトランザクションを取得するポートのベー

スクラスです。 

scl_nbget_server_t 
ノンブロッキング方式でトランザクションを準備するポートのベー

スクラスです。 

scl_nbput_port_t 
ノンブロッキング方式でトランザクションを送信するポートのベー

スクラスです。 

scl_nbput_server_t 
ノンブロッキング方式で送信されたトランザクションを処理するポ

ートのベースクラスです。 

scl_pass_port_t 一対 N のトランザクション処理をするポートのベースクラスです。 

scl_send_port_t 一対 N のトランザクションを転送するポートのベースクラスです。 

scl_receive_port_t 一対 N のトランザクションを受信するポートのベースクラスです。 
 

3.1.1 scl_port_t 

scl_port_t は以下のように定義されています。 

 

class scl_port_t #(type PARENT=scl_component_t) 

 extends scl_object_t; 

`scl_decl_object_m(scl_port_t#(PARENT)) 

// members 

PARENT     m_parent; 

scl_port_option_e m_option; 

 

// methods 

extern function new(string name,PARENT parent); 

port 

io_port 

pass_port 

get_port get_server put_port 

put_server 

send_port 

receive_port 

nbio_port 

nbget_port nbget_server nbput_port nbput_server 

nbput_server を接続する。 

 クライアントがトランザクションをノンブロッキング方式で出力するためには、put()

メソッドの代わりに try_put()メソッドを使用する。 

 

put()メソッドと殆ど同様に使用できるので、使用例を省略します。 

3.7 FIFO 

トランザクションを必要とする consumerとトランザクションを生成する producerが独立

プロセスとして実行するためには、トランザクションをプーリングして置く必要があります。

その際に使用できる FIFO キューが scl_fifo_t です。FIFO を使用すると、FIFO がトラン

ザクションの操作を行うのでサーバーになり、producer と consumer の何れもクライアン

トになります（図 3-13）。トランザクションの書き込みと取得に get()、try_get()、

put()、try_put()等を使用できます。つまり、ブロッキング方式およびノンブロッキング

方式の何れの方式でも FIFO にアクセスできます。 

 

 

図 3-13 FIFO を使用したトランザクションの授受 

 

このトランザクション処理法において、FIFO キューにトランザクションが存在すれば、

get()は直ぐに戻りますが、キューが空であれば、トランザクションの準備ができるまで

get()はブロックします。同様に、キューに空きがあれば、put()は直ぐに完了しますが、

キューが一杯であれば、キューに空きができるまで put()はブロックします。これらの一連

の制御を scl_fifo_t が行います。ここで、もし FIFO キューのサイズを無制限に設定して

おくと、常に put()は直ぐに戻ります。scl_fifo_t クラスのコンストラクタには FIFO キ

ューのサイズを指定する事ができるので、ユーザは FIFOキューの管理に柔軟性を持たせる事

ができます。 

 

なお、get()の代わりに try_get()を使用すれば、FIFO キューにトランザクションがあっ

てもなくてもすぐに戻ります。put()の代わりにtry_put()を使用しても同様な事が言えま

す。 

3.7.1 scl_fifo_t 

FIFOキューは以下のように定義されています。FIFOを使用するためには、トランザクション

のタイプを TR に指定しなければなりません。 

 

class scl_fifo_t 

 #(type TR=scl_transaction_t) extends scl_fifo_base_t#(TR); 

`scl_decl_component_m(scl_fifo_t#(TR)) 

// members 

 

// methods 

extern function 

 new(string name,scl_component_t parent=null,int size=0); 

extern virtual task put(input TR item); 

extern virtual function bit try_put(input TR item); 

extern virtual task get(output TR item); 

extern virtual function bit try_get(output TR item); 

extern virtual function int fifo_size(); 

extern virtual function int num_items(); 

extern virtual function bit is_empty(); 

extern virtual function bit is_full(); 

FIFO 

(scl_fifo_t) 
consumer 

put get 

producer 

try_get try_put 

extern virtual function bit can_put(); 

extern virtual function bit can_get(); 

extern virtual function void empty(); 

endclass 

 

FIFO で使用する TLM ポートやキューの定義は、ベースクラス scl_fifo_base_t で以下の

ように行われています。 

 

class scl_fifo_base_t 

 #(type TR=scl_transaction_t) extends scl_component_t; 

`scl_decl_component_m(scl_fifo_base_t#(TR)) 

// members 

scl_get_server_t#(TR,scl_fifo_base_t) m_get_server; 

scl_nbget_server_t#(TR,scl_fifo_base_t) m_nbget_server; 

scl_put_server_t#(TR,scl_fifo_base_t) m_put_server; 

scl_nbput_server_t#(TR,scl_fifo_base_t) m_nbput_server; 

mailbox   m_fifo; 

int     m_size; 

int     m_pending_requests; 

 

// methods 

extern function new(string name,scl_component_t parent=null,int 

size=0); 

extern virtual task put(input TR item); 

extern virtual function bit try_put(input TR item); 

extern virtual task get(output TR item); 

extern virtual function bit try_get(output TR item); 

endclass 

 

ベースクラスに定義されている重要なプロパーティを表 3-30 に示します。 

 

表 3-30 scl_fifo_base_t のプロパーティ 

プロパーティ 説明 

m_get_server 

ブロッキング方式でトランザクションを取得する場合に使用され

るポートです。scl_get_port_t タイプのポートをこのポートに接続

しなければなりません。 

m_nbget_server 

ノンブロッキング方式でトランザクションを取得する場合に使用

されるポートです。scl_nbget_port_tタイプのポートをこのポート

に接続しなければなりません。 

m_put_server 

ブロッキング方式でトランザクションを出力する場合に使用され

るポートです。scl_put_port_t タイプのポートをこのポートに接続

しなければなりません。 

m_nbput_server 

ノンブロッキング方式でトランザクションを出力する場合に使用

されるポートです。scl_nbput_port_t タイプのポートをこのポー

トに接続しなければなりません。 

m_fifo 
SystemVerilog の mailbox で実現されているキューです。トラン

ザクションはこのキューに登録されて管理されます。 

m_size 
FIFOキューのサイズを示します。m_size==0であれば、キューは

無制限の大きさを持ちます。 

m_pending_requests 

ブロッキング方式で get()を使用する場合、現在待ち状態にあるプ

ロセスの数を示しています。つまり、get()メソッドの処理が未完

了のプロセス数を示しています。 
 

FIFO で使用できるメソッドを表 3-31 に紹介します。 

4 virtual インターフェースの設定と取得 

virtualインターフェースは、検証コンポーネントが DUTをドライブする時や DUTからのレス

ポンスをサンプリングする際に使用される重要な手段です（図 4-1）。 

 

 

図 4-1 検証環境と virtual インターフェース 

 

DUT ではインターフェースを使用していなくても、virtual インターフェースに使用されてい

るインターフェースには DUT のポートが接続されているので、ドライバーやコレクターは恰

もDUTと直接接続されているような状況となります。つまり、DUTには恰もインターフェー

スが使用されているような状態となります。図 4-1 はその様な拡張解釈を基にしています。 

 

一般的には、柔軟性のある virtual インターフェースの設定手順は複雑になる傾向があります

が、SCL ではマクロを使用して使用手順の簡素化を実現しています。手順は以下のようにな

ります。 

 

 `scl_vif_config_m マクロを使用して virtualインターフェースを操作するためのクラ

スを定義する。 

 定義したクラスの set()メソッドを使用して virtual インターフェースを設定する。 

 定義したクラスの get()メソッドを使用して virtual インターフェースを取得する。 

 

本章では、以下のようなインターフェースが定義されていると仮定して、virtual インターフ

ェースの設定法と取得法を解説します。ここでは、インターフェースの内容は重要ではない

ので、省略しています。 

 

interface simple_if; 

... 

endinterface 

 

4.1 virtual インターフェースの使用準備 

以下に示すマクロを使用して virtual インターフェースを操作するためのクラスを定義します。

マクロに指定するパラメータの意味は、 

表 4-1 の通りです。 

 

`scl_vif_config_m(VIF_TYPE,VIF_CONFIG_NAME) 

 

表 4-1 `scl_vif_config_m マクロに指定するパラメータ 

パラメータ 意味 

VIF_TYPE 

定義すべき virtual インターフェースを指定します。このパラメータ

はユーザが使用するインターフェースの名称を指定します。ただし、

キーワード virtual を添えなければなりません。 

driver DUT collector 

TEST 

vif vif 

5 トランザクション 

本章では、トランザクションの定義法を解説します。トランザクションはオブジェクトであ

り、検証コンポーネントと多くの点で定義法が異なります。例えば、トランザクションは、

一般的に、階層を作らないので定義項目は DUT に関連するデータ項目に限定されます。 

5.1 scl_transaction_t 

トランザクションは、直接または間接的にscl_transaction_tのサブクラスとして定義し

なければなりません。以下のように、トランザクションのベースクラスが定義されています。 

 

class scl_transaction_t extends scl_object_t; 

`scl_decl_object_m(scl_transaction_t) 

 

extern function new(string name="transation"); 

extern function string get_type_name(); 

endclass 

 

`scl_decl_object_m マクロは、scl_transaction_t のクラスのタイプ情報を登録する

マクロで、以下の情報を生成します。 

 

static string m_type_name = "scl_transaction_t";  

function string get_type_name(); 

 get_type_name = m_type_name; 

endfunction 

 

ベースクラスに定義されている主なプロパーティは、表 5-1 のようになります。 

 

表 5-1 scl_transaction_t クラスの主なプロパーティ 

パラメータ 意味 

m_type_name トランザクションタイプを示す文字列です。 
 

参考 5-1 

ベースクラスの scl_transaction_t にはトランザクションのデータを表現する項目が定

義されていないので、ユーザはサブクラス内に必要なデータ項目を定義しなければなりませ

ん。つまり、scl_transaction_t をそのまま使用しても意味のあるトランザクション処理

を実現できません。 

□ 

 

ベースクラスに定義されているメソッドを以下に紹介します。 

5.1.1 new 

 

function new(string name="transation"); 

 

このメソッドは、トランザクションのインスタンスを作るコンストラクタを意味します。こ

のコンストラクタには、インスタンスを命名するための文字列を name に指定しなければな

りません。 

5.1.2 get_type_name 

 

function string get_type_name(); 

 

このトランザクションタイプの名称を文字列として戻します。サブクラスでは、このメソッ

ドを再定義しなければなりません。 

5.2 トランザクションの定義 

ユーザがトランザクションを定義するためには、以下に示すルールに従う必要があります。 

 

 トランザクションは scl_transaction_t またはそのサブクラスから定義されなけれ

ばなりません。 

 コンストラクタを必ず指定します。便利なマクロが準備されているので、それを使用す

るとコンストラクタの定義は簡単になります。 

 オブジェクトである事を示すためのマクロ `scl_object_m を指定します。 

 トランザクションにデータ項目がある場合には、フィールドマクロを指定してデータ項

目の属性を宣言する。このマクロを使用しないと、プリント処理等での対象には含まれ

まれなくなります。 

 

次に、トランザクションの簡単な定義例を紹介します。以下の例ではフィールドマクロを指

定していますが、第 6 章で解説されます。 

 

例 5-1 トランザクションの定義例 

トランザクションは、ベースクラスのscl_transaction_tを指定して以下のように定義し

ます。定義したトランザクションのクラスを SCL に登録するために ̀ scl_object_m マクロ

を指定しなければなりません。そして、トランザクション内に定義されているプロパーティ

に対しては、フィールドマクロを指定しなければなりません。 

 

class simple_item_t extends scl_transaction_t; 

logic [UP_DOWN_WIDTH-1:0] d, q, qn; 

logic    reset, load, up_down; 

 

 

`scl_object_field_begin_m(simple_item_t) 

 `scl_field_integral_m(reset,SCL_DEFAULT) 

 `scl_field_integral_m(load,SCL_DEFAULT) 

 `scl_field_integral_m(up_down,SCL_DEFAULT) 

 `scl_field_integral_m(d,SCL_DEFAULT) 

 `scl_field_integral_m(q,SCL_DEFAULT) 

 `scl_field_integral_m(qn,SCL_DEFAULT) 

`scl_field_end_m 

 

function new(string name="simple_item"); 

 super.new(name); 

endfunction 

 

endclass 

 

このように定義されたsimple_item_tのオブジェクトをprint()メソッドでプリントする

と以下のような書式でプリントされます。 

 

=========================================== 

Name         Type             Size    Value 

------------------------------------------- 

item         simple_item_t    -       @16   

  reset      integral         1       'h0   

  load       integral         1       'h0   

  up_down    integral         1       'h0   

  d          integral         8       'h00  

  q          integral         8       'ha3  

  qn         integral         8       'h5c 

=========================================== 

 

フィールドマクロ 

トランザクション

のベースクラスを

指定する 

コンストラクタの定義 

6 フィールドマクロ 

トランザクションのデータ項目にはフィールドマクロを使用して属性を指定しなければなり

ません。フィールドマクロを指定するとトランザクションをプリントする機能が自動生成さ

れます。 

 

トランザクションにデータ項目が存在する事を SCL に知らせるためには、以下のように

`scl_field_begin_m と `scl_field_end_m マ ク ロ で 囲 み ま す 。 そ の 間 に 、

`scl_field_*_m マクロを使用してデータ項目を指定します。 

 

`scl_field_begin_m(CLASS_TYPE) 

... 

`scl_field_end_m 

 

参考 6-1 

トランザクションにデータ項目が存在しない場合には、`scl_field_begin_m と

`scl_field_end_m マクロを指定する必要はありません。 

□ 

 

フィールドマクロの一般形式は、以下のようになります。ここで、VAR はデータ項目を示す

変数名で、OPTION はデータ項目に関する取扱い方法を示します。 

 

`scl_field_type_m(VAR,OPTION) 

 

データ項目が enum型である場合には、VARの直前に enum型を指定しなければなりません。

OPTION としては、現在はプリント処理機能のみが実装されているため、SCL_DEFAULT を

指定しておけば十分です。現在実装されている OPTION を表 6-1 に示します。 

 

表 6-1 OPTION の指定項目 

OPTION 機能 

SCL_UNSIGNED 
byte、shortint、int、longint、time、integer 型の値を符号なし

として扱うためのオプションです。 

SCL_BREAKUP 数値の桁を考慮して見易くプリントする機能です。 

SCL_COPYABLE 

該当するフィールドをコピー処理の対象にするオプションです。

ブロッキング代入文で割り当て可能なフィールドに対して、この

オプションを指定して下さい。クラスのハンドルに関しては、

shallow copy が適用されます。deep copy をする場合には、ユーザ

自身でコピー処理を準備して下さい。 

SCL_DEFAULT 標準値を示します。SCL_COPYABLE は標準値に含まれます。 

 

参考 6-2 

SCL_DEFAULT からオプションを外すためには、 SCL_DEFAULT&~option として下さい。例

えば、以下のようにすると、SCL_DEFAULT から SCL_COPYABLE が外されます。 
 

`scl_field_integral_m(d,SCL_DEFAULT&~SCL_COPYABLE) 

 

□ 

 

フィールドマクロの開始と終了の意味は、表 6-2 の通りです。 

表 6-2 フィールドマクロの開始と終了 

フィールドマクロ名 機能 

`scl_field_begin_m(CLASS_TYPE) 
フ ィ ー ル ド マ ク ロ の 開 始 を 示 し ま す 。

CLASS_TYPE にはクラス名を指定して下さい。 

`scl_field_end_m フィールドマクロの終了を示します。 
 

オブジェクトやコンポーネントの定義をする際に指定するマクロとフィールド開始を宣言す

るマクロを一つにまとめるマクロもあります（表 6-3）。 

 

表 6-3 短縮形のフィールド開始マクロ 

基本的な使用法 短縮形マクロ 

`scl_object_m(TYPE) 
`scl_field_begin_m(TYPE) 
… 
`scl_field_end_m 

`scl_object_field_begin_m(TYPE) 
… 
`scl_field_end_m 

`scl_component_m(TYPE)  
`scl_field_begin_m(TYPE) 
… 
`scl_field_end_m 

`scl_component_field_begin_m(TYPE) 
… 
`scl_field_end_m 

 
例えば、オブジェクトを定義する際には以下のようにマクロを使用します。 
 

`scl_object_m(simple_item_t) 

`scl_field_begin_m(simple_item_t) 

 `scl_field_integral_m(a,SCL_DEFAULT) 

 ... 

`scl_field_end_m 

 
こうする代わりに、以下のように短縮形を指定できます。 
 

`scl_object_field_begin_m(simple_item_t) 

 `scl_field_integral_m(a,SCL_DEFAULT) 

 ... 

`scl_field_end_m 

 
使用するフィールドマクロは、変数が単数形であるか、あるいはアレイであるかで大きく異

なります。以下では、それぞれを区別してマクロの種類と意味を解説します。 

6.1 単数形変数 

変数が単数形である場合には、SystemVerilog のデータタイプに対応したフィールドマクロが

定義されています。 

6.1.1 `scl_field_integral_m 

 

` scl_field_integral_m(VAR,OPTION) 

 

logic、bit、reg 型の変数を指定する場合に使用します。 

6.1.2 ` scl_field_byte_m 

 

` scl_field_byte_m(VAR,OPTION) 

 

byte 型の変数に使用します。 

7 検証コンポーネント 

検証コンポーネントを開発するためのベースクラスとしては、表 7-1 に示すクラスがありま

す。これらのクラスは、メソドロジークラスと呼ばれ、scl_component_t クラスを基にし

て作られています。本章では、これらの検証コンポーネントを解説します。 

  

表 7-1 主要な検証コンポーネントのクラス 

SCL クラス 機能 

scl_driver_t 

ドライバーのベースクラスです。ユーザはドライバーをこのクラスの

サブクラスとして定義しなければなりません。ドライバーは、ジェネ

レータからトランザクションを取得する機能を備えています。 

scl_generator_t 

ジェネレータのベースクラスです。ジェネレータは、ドライバーの要

求に応じてトランザクションを生成して戻します。実際には、シナリ

オにより割り当てられたプレイを介して、トランザクションを生成し

て貰います。 

scl_collector_t 

コレクターのベースクラスです。コレクターは、DUTからのレスポン

スをサンプリングしてトランザクションに変換して、モニターにトラ

ンザクションを送信する役目を持ちます。一般的には、コレクターは

検証に関わる作業を担当しません。 

scl_monitor_t 

モニターのベースクラスです。モニターは、コレクターから受信した

トランザクションを他の検証コンポーネントに一斉に転送します。モ

ニター自身も簡単な検証作業を行います。 

scl_agent_t 

エージェントのベースクラスです。エージェントは、ドライバー、ジ

ェネレータ、コレクター、モニターから構成される最小単位の階層的

検証コンポーネントです。 

scl_env_t 

エンバイロンメントのベースクラスです。エンバイロンメントは、エ

ージェントや他のエンバイロンメントから構成される階層的な検証コ

ンポーネントで、検証内容に応じて様々な規模のエンバイロンメント

が開発されます 

scl_scoreboard_t 
スコアボードのベースクラスです。スコアボードは、DUTからのレス

ポンスを検証する役目を持ちます。 

scl_test_t 

テストのベースクラスです。テストは、一般に、複数のエンバイロン

メントから構成されますが、テスト同士が共有する資源をベースクラ

スとして定義するのが一般的です。 

 

以下では、検証コンポーネントのベースクラスの仕様と機能を解説し、ユーザの検証コンポ

ーネントを定義する手順を紹介します。 

7.1 ドライバー 

ドバイバーは、get ポートを使用してジェネレータからトランザクションを取得しますが、

scl_driver_t クラスはトランザクションを操作するための基本的機能を提供します。ユー

ザは、このクラスからサブクラスを定義する事により、定型的な準備作業から解放されます。 

7.1.1 ドライバーのベースクラス 

ドライバーのベースクラス scl_driver_t は以下のように定義されています。 

 

class scl_driver_t #(type TR=scl_transaction_t) 

 extends scl_component_t; 

`scl_decl_component_m(scl_driver_t#(TR)) 

// members 

scl_get_port_t#(TR,scl_driver_t)  m_get_port; 

 

// methods 

extern function new(string name,scl_component_t parent); 

8 検証手順とシナリオ 

シナリオとは、テストケースを実行するための手順で、バッチ処理のスクリプトの役割に相

当します。テストケースがシナリオを決定すると、ジェネレータにプレイ（脚本）が割り当

てられ、トランザクションを生成するための脚本が確定します。選択された脚本は、幾つか

のアクト（幕）から構成されて、脚本に記されている順に幕が進行します。このようにして、

検証手順が遂行されて行きます（図 8-1）。本章では、テストを構成するジェネレータ、シナ

リオ、プレイ、アクトの関係を詳しく解説します。 

 

 

図 8-1 シナリオの役割とトランザクション生成処理 

8.1 シナリオ 

シナリオはオブジェクトですが、特別なインスタンスを作らずに使用できます。したがって、

ユーザはシナリオのサブクラスを定義する必要がないので、使用法は非常に簡単です。 

8.1.1 シナリオのベースクラス 

シナリオのベースクラスは、以下のように定義されています。 

 

class scl_scenario_t extends scl_object_t; 

typedef scl_config_t#(scl_play_t) CONFIG_TYPE; 

// members 

const static string m_type_name = "scl_scenario_t"; 

 

// methods 

extern function new(string name="scenario"); 

extern function string get_type_name(); 

extern static function void 

 allocate(input string path,scl_play_t play); 

extern static function bit 

 get(input string path,output scl_play_t play); 

endclass 

 

主なプロパーティは、表 8-1 にまとめられています。 

テスト 

ジェネレータ 

シナリオ 

プレイ 

アクト アクト アクト 

トランザ

クション 

トランザ

クション 

トランザ

クション 

テストはプレ

イのインスタ

ンスを作り、

シナリオを呼

び出す 

シナリオはジェネレータと

プレイを結びつける 

  `scl_create_object_m(play_reset_down_t,"play")); 

endfunction 

endclass 

 

■ 

 

例 8-2 ジェネレータに割り当てられているプレイを取得する例 

ジェネレータのベースクラスは connect_phase()でシナリオを利用して割り当てられてい

るプレイを取得します。 

 

class scl_generator_t #(type TR=scl_transaction_t) 

 extends scl_component_t; 

... 

function void connect_phase(scl_run_param_t param); 

 if( !scl_scenario_t::get(get_full_name(),m_play) ) 

  scl_warning({get_full_name(), 

   ": ",scl_global_t::m_msg_couldnt_get_default_scenario}); 

endfunction 

endclass 

 

■ 

 

参考 8-1 

scl_scenario_t::get()は検証コンポーネントに割り当てられたプレイを取得しますが、

この取得処理は scl_generator_t クラスが行うため、ユーザがこのメソッドを使用する必

要はありません。しかし、既に解説したように、ユーザがシナリオをテストケースに定義す

るためには scl_scenario_t:: allocate()メソッドを使用しなければなりません。 

□ 

 

8.2 プレイ 

プレイはオブジェクトであり、トランザクションを作る手順を定義する機能を持ちます。プ

レイは、仮想的にアクトのリストを保有しています。シミュレーション開始時には、リスト

の最初のアクトがアクティブでありそれが実行されると、リスト上にある次のアクトがアク

ティブになります。こうして、リストの最初から最後までのアクトが順に実行されるように

なります。 

 

ドライバーがジェネレータにトランザクションを要求すると、ジェネレータは現在アクティ

ブなアクトを実行してトランザクションを生成して貰います。その後、ジェネレータは生成

されたトランザクションをドライバーに戻します。 

8.2.1 プレイのベースクラス 

プレイのベースクラスは、以下のように定義されています。パラメータとしてトランザクシ

ョンのタイプを指定しなければなりません。 

 

class scl_play_t #(type TR=scl_transaction_t) extends scl_object_t; 

`scl_decl_object_m(scl_play_t#(TR)) 

// members 

TR      m_item; 

scl_item_status_e m_item_status; 

bit      m_reuse_transaction; 

 

// methods 

extern function new(string name); 

ここで使用したマクロ`scl_act_get_item_m と`scl_act_get_group_m は 8.3.3 項で解

説されます。 

■ 
 

8.3 アクト 

アクトは、トランザクションを作る機能を持ちます。原則として、一つのアクトは一つのト

ランザクションを作ると役目を終わりますが、アクトに階層を設けて複数のアクトを生成す

る事もできます。 

8.3.1 アクトのベースクラス 

アクトのベースクラスは、以下のように定義されています。パラメータとしてトランザクシ

ョンのタイプを指定しなければなりません。 

 

class scl_act_t #(type TR=scl_transaction_t) extends scl_object_t; 

`scl_decl_object_m(scl_act_t#(TR)) 

typedef scl_play_t#(TR)  PLAY_TYPE; 

// members 

PLAY_TYPE  m_play; 

 

// methods 

extern function new(string name); 

extern function void set_play(PLAY_TYPE play); 

extern virtual task get_item(); 

extern virtual task get_group(); 

extern virtual function void make_item(); 

endclass 

 

`scl_decl_object_mマクロは、scl_act_tのクラスのタイプ情報を登録するマクロで、

以下の情報を生成します。 

 

static string m_type_name = "scl_act_t#(TR)";  

function string get_type_name(); 

 get_type_name = m_type_name; 

endfunction 

 

主なプロパーティは、表 8-4 にまとめられています。 

 

表 8-4 scl_act_t の主なプロパーティ 

プロパーティ 説明 

m_type_name 
このクラスタイプを示す文字列です。サブクラスにより再定義さ

れます。 

m_play このアクトが属するプレイを示すハンドルです。 
 

参考 8-2 

アクトを定義する際、get_group()メソッドか make_item()メソッドのいずれかを選択す

る事になります。両方のメソッドを定義する必要はありません。アクトがツリーのリーフノ

ードを表現するのであれば、make_item()メソッドを定義し、アクトがツリーの内部ノー

ドを表現するのであれば、get_group()メソッドを定義すれば良いです。 

□ 

 

このクラスに定義されている主なメソッドを以下にまとめます。 



検証のための SystemVerilog クラスライブラリー 

120 

Copyright 2024 © Artgraphics. All rights reserved. 

 

図 8-4 シミュレーション実行開始直後のツリーの状態 

 

ドライバーがトランザクションの要求をすると、ジェネレータはアクティブなアクトからト

ランザクションを受け取りドライバーに戻します。同時に、プレイの get_item()は次のア

クティブノードを作ります（図 8-4 (b)）。このようにして、トランザクションのツリーが順

に完成して行きます。シミュレーションが終了した時にはツリーが完成した事になります。 

 

ツリーのルートノードはプレイで、リーフノードがトランザクションを作り、内部ノードは

階層を作るための役割を持ちます。リーフノードは左から順に実行され、そのうちの一つだ

けがアクティブなノードで、アクティブなノードがトランザクションを生成します。内部ノ

ードは、アクトの get_group()メソッドに対応し、リーフノードはアクトの make_item()

メソッドに対応します。最終的には、図 8-5 に示すようなツリーが完成します。 

 

 

図 8-5 トランザクション作成用のツリー 

参考 8-3 

本書でトランザクション生成ツリーを図示する時には完成したツリーで表現しますが、実際

には一部のノードや枝は未完成の状態にあると理解して下さい。 

□ 
 

8.4.3 ジェネレータとアクトのハンドシェーク 

ドライバーが、ジェネレータにトランザクションを要求すると、ジェネレータがアクトに対

して REQUEST_ITEM イベントを起こします。すると、現在アクティブなアクトがトランザ

クションを作り ITEM_READY イベント起こします。そして、ジェネレータは作られたトラ

ンザクションをドライバーに戻します。同時に、現在アクティブであったノードの次に並ん

でいるリーフノードが次のアクティブなノードになります。全てのリーフノードの処理が終

了するとトランザクション処理の終了となります。これらの流れを図 8-6 に示します。尚、

既に紹介したように内部ノードを利用するとリーフノードを何回か繰り返す事もできます。 

 

プレイ 

(a) (b) 

内 部 ノ ー ド は
get_group()

メソッドに対応 
リーフノードは
make_item()

メソッドに対応 

アクティブなノード 

プレイ 

図 8-6 トランザクション生成処理の流れ 

  

ドライバー ジェネレータ 

 

REQUEST_ITEM 

ITEM_READY 

get(item) 

item 

トランザクション 

処理プロセス 

9 検証環境構築例 

本章では検証環境に SCL を適用する例を紹介します。簡単なシーケンシャル回路（非同期リ

セット信号を持つアップダウンカウンター）を DUT として検証環境を構築します。以下では、

二種類の検証環境構築法を紹介します。最初の検証環境では、モニターが検証結果をプリン

トしますが、二番目の検証環境では、スコアボードを使用して検証結果を確認します。 

9.1 検証環境（モニターによる検証） 

図 9-1 に示すような検証環境を構築します。また、表 9-1 は検証環境の主な構成要素の機能を

示しています。 

 

 

図 9-1 SCL による検証環境構築例 

 

表 9-1 検証環境を構成する主な要素 

構成要素 機能 

up_down_counter DUT のアップダウンカウンターです。 

pkg_definitions 検証環境で共有する情報を定義しておきます。 

simple_if インターフェースを示します。 

simple_item_t トランザクションを示すクラスです。 

act_down_t 

トランザクションは RESET、LOAD、UP、DOWN 等のコマンド

機能を備えているので、それらのコマンドに対応するアクトを定

義しておきます。 

act_load_t 
act_loaddown_t 
act_reset_t 
act_up_t 
act_upup_t 
act_upuploaddown_t 
play_t テストケースのベースクラスです。 

play_reset_down_t テストケース 1 のシナリオを生成するクラスです。 

play_reset_up_t テストケース 2 のシナリオを生成するクラスです。 

driver_t DUT をドライブするドライバーのクラスです。 

generator_t 

ドライバーの要求によりトランザクションを生成するクラスで

す。このクラスの run_phase()には、テストケースのシナリオが

割り当てられます。 

collector_t 
DUT からのレスポンスをサンプリングするコレクターのクラス

です。 

monitor_t コレクターから受信したトランザクションの処理をします。この

DUT 

env 

 

test 

driver generator 

monitor collector 

vif 

vif 

agent 

例では、トランザクションをプリントするだけの簡単な処理内容

になっています。 

agent_t 
ドライバー、シーケンサー、コレクター、モニターから構成され

るエージェントのクラスです。 

env 
エージェントから構成されるトップレベルのエンバイロンメント

クラスです。 

test_base_t 
テストのベースクラスを示します。二つのテストに共通する機能

をベースクラスに吸収し、テストの記述を簡略化します。 

test1_t テストケース 1 を実行するクラスです。 

test2_t テストケース 2 を実行するクラスです。 

pkg 検証環境に必要な資源をパッケージに集めておきます。 

top トップモジュールを示します。 

 

検証環境の構成要素を順に解説します。 

9.1.1 up_down_counter 

アップダウンカウンターは、非同期なリセット信号を持ち、通常の動作時には load 信号と

up_down 信号の値により機能が決定されます（表 9-2）。load 信号の方が up_down 信号よ

りも優先順位が高く設定されています。 

 

表 9-2 アップダウンカウンターの仕様 

ポート 意味 

clk クロック信号です。 

reset 非同期なリセット信号です。 

load load==1 であれば、データを外部からロードします。 

up_down 

load==0 の時に動作する機能で、up_down==1 であればカウンタ

ーをインクリメントし、up_down==0であればカウンターをデク

リメントします。 

d load==1 の時にレジスターにロードする値を意味します。 

q 現在のカウンター値を示します。 

qn q の値をビット毎に反転した値を示します。 
 

アップダウンカウンターの記述は以下のようになります。 

 

module up_down_counter #(NBITS=4) 

 (input clk,reset,load,up_down,logic [NBITS-1:0] d, 

 output logic [NBITS-1:0] q,qn); 

logic [NBITS-1:0] counter; 

 

assign q = counter; 

assign qn = ~counter; 

 

always @(posedge clk,posedge reset) 

 if( reset ) 

  counter <= 0; 

 else if( load ) 

  counter <= d; 

 else if( up_down ) 

  counter <= counter + 1; 

 else 

  counter <= counter - 1; 

   

endmodule 

9.1.2 pkg_definitions 

以下のように、検証環境で共有される情報を定義しておきます。 

 

package pkg_definitions; 

parameter UP_DOWN_WIDTH = 8; 

typedef enum { RESET, LOAD, UP, DOWN } up_down_op_e; 

endpackage 

 

DUT を操作するためのコマンドは up_down_op_e に定義されていますが、コマンドは表 9-3

のような意味を持ちます。 

 

表 9-3 up_down_op_e の定義内容 

シンボル 意味 

RESET DUT をリセットします。 

LOAD DUT に値をロードさせます。 

UP DUT のカウンターをインクリメントさせます。 

DOWN DUT のカウンターをデクリメントさせます。 

 

9.1.3 simple_if 

インターフェースには、DUT に接続する信号とクロッキングブロックを定義しておきます。

そして、virtual インターフェースを操作するためのクラス（vif_config）を生成しておき

ます。 

 

interface simple_if import pkg_definitions::*; (input logic clk); 

logic [UP_DOWN_WIDTH-1:0] d, q, qn; 

logic       reset, load, up_down; 

 

clocking cb @(posedge clk); endclocking 

clocking cbr @(posedge reset); endclocking 

 

initial begin 

 reset = 0; 

 load = 0; 

 up_down = 1; 

 d = '0; 

end 

endinterface 

 

`scl_vif_config_m(virtual simple_if,vif_config) 

 

9.1.4 simple_item_t 

トランザクションには、DUT のポートに対応する変数を全て定義しておきます。 

 

class simple_item_t extends scl_transaction_t; 

logic [UP_DOWN_WIDTH-1:0] d, q, qn; 

logic       reset, load, up_down; 

`scl_object_field_begin_m(simple_item_t) 

 `scl_field_integral_m(reset,SCL_DEFAULT) 

 `scl_field_integral_m(load,SCL_DEFAULT) 

 `scl_field_integral_m(up_down,SCL_DEFAULT) 

 `scl_field_integral_m(d,SCL_DEFAULT) 

 `scl_field_integral_m(q,SCL_DEFAULT) 

 `scl_field_integral_m(qn,SCL_DEFAULT) 

`scl_field_end_m 

virtual インターフェースを操作す

るためのクラスを生成する 

`scl_object_new_default_m("simple_item") 

extern function up_down_op_e make_op(); 

endclass 

 

make_op()メソッドは、reset、load、up_down からコマンド RESET、LOAD、UP、

DOWN を作り出します。このメソッドは、以下のように定義されています。 

 

function up_down_op_e simple_item_t::make_op(); 

 if( reset ) 

  make_op = RESET; 

 else if( load ) 

  make_op = LOAD; 

 else if( up_down ) 

  make_op = UP; 

 else 

  make_op = DOWN; 

endfunction 

 

9.1.5 act_down_t 

DUT に DOWN コマンドを送るためのアクトです。 

 

class act_down_t extends scl_act_t#(simple_item_t); 

`scl_object_m(act_down_t) 

`scl_object_new_m 

extern function void make_item(); 

endclass 

 

// make_item 

function void act_down_t::make_item(); 

 m_play.m_item.reset = 0; 

 m_play.m_item.d = 0; 

 m_play.m_item.up_down = 0; 

 m_play.m_item.load = 0; 

endfunction 

 

9.1.6 act_load_t 

DUT に LOAD コマンドを送るためのアクトです。 

 

class act_load_t extends scl_act_t#(simple_item_t); 

`scl_object_m(act_load_t) 

`scl_object_new_m 

extern function void make_item(); 

endclass 

 

// make_item 

function void act_load_t::make_item(); 

 m_play.m_item.reset = 0; 

 m_play.m_item.d = $random; 

 m_play.m_item.up_down = 0; 

 m_play.m_item.load = 1; 

endfunction 

 

9.1.7 act_loaddown_t 

階層的なアクトを表現し、LOAD と DOWN コマンドを順に生成します（図 9-2）。 

class act_loaddown_t extends scl_act_t#(simple_item_t); 

// members 

act_load_t  act_load; 

act_down_t  act_down; 

`scl_object_m(act_loaddown_t) 

`scl_object_new_m 

 

// methods 

task get_group(); 

 `scl_act_get_item_m(act_load,m_play) 

 `scl_act_get_item_m(act_down,m_play) 

endtask 

endclass 

 

 

図 9-2 act_loaddown_t が表現する部分ツリー 

 

9.1.8 act_reset_t 

 

DUT に RESET コマンドを送るためのアクトです。 

 

class act_reset_t extends scl_act_t#(simple_item_t); 

`scl_object_m(act_reset_t) 

`scl_object_new_m 

extern function void make_item(); 

endclass 

 

// make_item 

function void act_reset_t::make_item(); 

 m_play.m_item.reset = 1; 

 m_play.m_item.d = 0; 

 m_play.m_item.up_down = 0; 

 m_play.m_item.load = 0;  

endfunction 

 

9.1.9 act_up_t 

DUT に UP コマンドを送るためのアクトです。 

 

class act_up_t extends scl_act_t#(simple_item_t); 

`scl_object_m(act_up_t) 

`scl_object_new_m 

extern function void make_item(); 

endclass 

 

// make_item 

function void act_up_t::make_item(); 

 m_play.m_item.reset = 0; 

 m_play.m_item.d = 0; 

 m_play.m_item.up_down = 1; 

 m_play.m_item.load = 0; 

endfunction 

 

act_loaddown_t 

act_load_t act_down_t 

9.1.10 act_upup_t 

このアクトは、階層的なコマンドを作るための例として導入しました。この場合には、UP コ

マンドを二回繰り返すように get_group()タスクを定義しています。アクトに複雑な処理

がある場合には、このようにして処理の細分化を実現できます。このアクトは図 9-3 のよう

な部分ツリーを表現します。 

 

class act_upup_t extends scl_act_t#(simple_item_t); 

// members 

act_up_t  act_up; 

`scl_object_m(act_upup_t) 

`scl_object_new_m 

 

// methods 

task get_group(); 

 `scl_act_get_item_m(act_up,m_play); 

 `scl_act_get_item_m(act_up,m_play); 

endtask 

 

endclass 

 

 

図 9-3 act_upup_t が表現する部分ツリー 

 

9.1.11 act_upuploaddown_t 

階層的なアクトを表現しています。階層的な act_upup_t のアクトと act_loaddown_t の

アクトを順に呼び出します（図 9-4）。このように階層の深さに制限はありません。 

 

class act_upuploaddown_t extends scl_act_t#(simple_item_t); 

// members 

act_upup_t  act_upup; 

act_loaddown_t act_loaddown; 

`scl_object_m(act_upuploaddown_t) 

`scl_object_new_m 

 

// methods 

task get_group(); 

 `scl_act_get_group_m(act_upup,m_play); 

 `scl_act_get_group_m(act_loaddown,m_play); 

endtask 

endclass 

 

 

図 9-4 act_upuploaddown_t が表現する部分ツリー 

9.1.12 play_t 

テストケースのベースクラスを以下のように定義しておきます。役目は、それぞれのテスト

ケースが使用する変数を定義するだけです。 

act_upup_t 

act_up_t act_up_t 

act_upuploaddown_t 

act_upup_t act_loaddown_t 

class play_t extends scl_play_t#(simple_item_t); 

act_reset_t act_reset; 

act_down_t act_down; 

act_up_t  act_up; 

act_load_t act_load; 

act_upuploaddown_t act_upuploaddown; 

`scl_object_m(play_t) 

`scl_object_new_m 

endclass 

 

9.1.13 play_reset_down_t 

テストケース 1 のためのクラスです（図 9-5）。 

 

class play_reset_down_t extends play_t; 

// members 

// 

`scl_object_m(play_reset_down_t) 

// methods 

`scl_object_new_default_m("play_reset_down"); 

 

task get_item(); 

 `scl_act_get_item_m(act_reset,this) 

 `scl_act_get_group_m(act_upuploaddown,this) 

endtask 

endclass 

 

 

図 9-5 テストケース 1 のツリー 

9.1.14 play_reset_up_t 

テストケース 2 のクラスです。このテストケースでは階層を設けずにテストを実現していま

す（図 9-6）。 

 

class play_reset_up_t extends play_t; 

// members 

`scl_object_m(play_reset_up_t) 

 

// methods 

`scl_object_new_default_m("play_reset_up") 

 

task get_item(); 

 `scl_act_get_item_m(act_reset,this) 

 `scl_act_get_item_m(act_down,this) 

 `scl_act_get_item_m(act_load,this) 

 `scl_act_get_item_m(act_up,this) 

 `scl_act_get_item_m(act_up,this) 

play_reset_down_t 

act_reset_t act_upuploaddown_t 

act_upup_t act_loaddown_t 

act_up_t act_up_t 

act_load_t 

act_down_t 

階層を持たないアクト 

階層的なアクト 

endtask 

endclass 

 

 

図 9-6 テストケース 2 のツリー 

9.1.15 driver_t 

ドライバーの全容は以下のようになります。必ず、virtual インターフェースを宣言しなけれ

ばなりません。 

 

class driver_t extends scl_driver_t#(simple_item_t); 

vif_config::vif_type vif; 

`scl_component_m(driver_t) 

`scl_component_new_m 

`scl_extern_connect_phase_m 

`scl_extern_run_phase_m 

extern task drive_dut(TR item); 

endclass 

 

connect_phase()において、以下のようにして virtual インターフェースを設定します。 

 

`scl_connect_phase_m(driver_t) 

 `scl_super_connect_phase_m 

 vif = vif_config::get(); 

`scl_end_connect_phase_m 

 

run_phase()では以下のようにして、ジェネレータよりトランザクションを取得して、

DUT をドライブします。 

 

`scl_run_phase_m(driver_t) 

TR item; 

 m_get_port.m_connected_port.set_option( 

   SCL_PORT_REUSE_TRANSACTION); 

 forever begin 

  m_get_port.get(item); 

  drive_dut(item); 

  @(negedge vif.clk); 

 end 

`scl_end_run_phase_m 

 

DUT をドライブする部分は以下のようになります。 

 

task driver_t::drive_dut(TR item); 

 vif.reset <= item.reset; 

 vif.load <= item.load; 

 vif.up_down <= item.up_down; 

 vif.d <= item.d; 

 if( item.reset ) 

  vif.reset = #1 0; 

endtask 

play_reset_up_t 

act_reset_t act_down_t act_load_t act_up_t act_up_t 

virtual インターフェースの定義 

参考 9-1 

非同期信号 reset に値を設定する場合には、ノンブロッキング代入文を使用すると安全です。

一般的に、シーケンシャル回路を検証する際に DUT をドライブするためには、ノンブロッキ

ング代入文を使用する方が安全です。 

□ 

 

9.1.16 generator_t 

ジェネレータは、ベースクラスが殆どの処理を担当するので、ユーザは以下のようにジェネ

レータを定義するだけで済みます。 

 

`scl_generator_m(generator_t,simple_item_t) 

 

9.1.17 collector_t 

コレクターの全容は以下のようになります。ドライバーと同様に virtual インターフェースを

宣言しなければなりません。 

 

class collector_t extends scl_collector_t#(simple_item_t); 

vif_config::vif_type vif; 

simple_item_t  item; 

`scl_component_m(collector_t) 

`scl_component_new_m 

`scl_extern_connect_phase_m 

`scl_extern_run_phase_m 

extern task collect_regular(); 

extern task collect_reset(); 

extern function void send_item(); 

endclass 

 

connect_phase()で virtual インターフェースの使用を以下のように準備します。 

 

`scl_connect_phase_m(collector_t) 

 vif = vif_config::get(); 

`scl_end_connect_phase_m 

 

run_phase()では、DUT からのレスポンスを監視するために、クロッキングブロックを使

用してイベントの発生を待ちます。 

 

`scl_run_phase_m(collector_t) 

 item = `scl_create_object_m(simple_item_t,"item"); 

 fork 

  collect_regular(); 

  collect_reset(); 

 join 

`scl_end_run_phase_m 

 

クロックのイベントを collect_regular()メソッドで監視します。 

 

task collector_t::collect_regular(); 

 forever begin 

  @vif.cb; 

  send_item(); 

 end 

endtask 

virtual インターフェースの定義 


